```
> restart;
> ynp1:=a0*y(tn)+a1*y(tn-h)+a2*y(tn-2*h)+
      h*(bm1*D(y)(tn+h)+b0*D(y)(tn)+b1*D(y)(tn-h)+
        b2*D(y)(tn-2*h))+E5*h^5*(D@@5)(y)(theta)/5!;
```

$$ynp1 := a0\,y(tn) + a1\,y(tn-h) + a2\,y(tn-2\,h) + h\,(bm1\,\mathrm{D}(y)(tn+h) + b0\,\mathrm{D}(y)(tn) + b1\,\mathrm{D}(y)(tn-h) + b2\,\mathrm{D}(y)(tn-2\,h)) + \frac{E5\,h^5\,\mathrm{D}^{(5)}(y)(\theta)}{120} \tag{1}$$

```
> r:=y(tn+h)-ynp1;
```

$$r := y(tn+h) - a0\,y(tn) - a1\,y(tn-h) - a2\,y(tn-2\,h) - h\,(bm1\,\mathrm{D}(y)(tn+h) + b0\,\mathrm{D}(y)(tn) + b1\,\mathrm{D}(y)(tn-h) + b2\,\mathrm{D}(y)(tn-2\,h)) - \frac{E5\,h^5\,\mathrm{D}^{(5)}(y)(\theta)}{120} \tag{2}$$

```
> eq[0]:=eval(subs(y=(x->1),r));
```

$$eq_0 := 1 - a0 - a1 - a2 \tag{3}$$

```
> for j from 1 to 5 do
     tmp[j]:=eval(subs(y=(x->x^j),r));
     eq[j]:=coeff(tmp[j],h^j);
     print(eq[j]);
  od:
```

$$1 + a1 + 2\,a2 - bm1 - b0 - b1 - b2$$
$$1 - a1 - 4\,a2 - 2\,bm1 + 2\,b1 + 4\,b2$$
$$1 + a1 + 8\,a2 - 3\,bm1 - 3\,b1 - 12\,b2$$
$$1 - a1 - 16\,a2 - 4\,bm1 + 4\,b1 + 32\,b2$$
$$1 + a1 + 32\,a2 - 5\,bm1 - 5\,b1 - 80\,b2 - E5 \tag{4}$$

```
> S1:=solve({seq(eq[k],k=0..5)},{a0,bm1,b0,b1,b2,E5});
```

$$S1 := \left\{E5 = -\frac{19}{6} + \frac{11\,a1}{6} - \frac{4\,a2}{3},\ a0 = 1 - a1 - a2,\ b0 = \frac{19}{24} + \frac{13\,a1}{24} + \frac{a2}{3},\ b1 = -\frac{5}{24} + \frac{13\,a1}{24} + \frac{4\,a2}{3},\ b2 = \frac{1}{24} - \frac{a1}{24} + \frac{a2}{3},\ bm1 = \frac{3}{8} - \frac{a1}{24}\right\} \tag{5}$$

```
> method:=subs(E5=0,ynp1);
```

$$method := a0\,y(tn) + a1\,y(tn-h) + a2\,y(tn-2\,h) + h\,(bm1\,\mathrm{D}(y)(tn+h) + b0\,\mathrm{D}(y)(tn) + b1\,\mathrm{D}(y)(tn-h) + b2\,\mathrm{D}(y)(tn-2\,h)) \tag{6}$$

```
> m2:=subs(S1,method);
```

$$m2 := (1 - a1 - a2)\,y(tn) + a1\,y(tn-h) + a2\,y(tn-2\,h) + h\left(\left(\frac{3}{8} - \frac{a1}{24}\right)\mathrm{D}(y)(tn+h) + \left(\frac{19}{24} + \frac{13\,a1}{24} + \frac{a2}{3}\right)\mathrm{D}(y)(tn) + \left(-\frac{5}{24} + \frac{13\,a1}{24} + \frac{4\,a2}{3}\right)\mathrm{D}(y)(tn-h) + \left(\frac{1}{24} - \frac{a1}{24} + \frac{a2}{3}\right)\mathrm{D}(y)(tn-2\,h)\right) \tag{7}$$

```
> m3:=eval(subs(D(y)=(x->f(x,y(x))),m2));
```

$$m3 := (1 - a1 - a2)\,y(tn) + a1\,y(tn-h) + a2\,y(tn-2\,h) + h\left(\left(\frac{3}{8} - \frac{a1}{24}\right)f(tn+h,\,y(tn+h)) + \left(\frac{19}{24} + \frac{13\,a1}{24} + \frac{a2}{3}\right)f(tn,\,y(tn)) + \left(-\frac{5}{24} + \frac{13\,a1}{24} + \frac{4\,a2}{3}\right)f(tn-h,\,y(tn-h)) + \left(\frac{1}{24} - \frac{a1}{24} + \frac{a2}{3}\right)f(tn-2\,h,\,y(tn-2\,h))\right) \tag{8}$$

```
> f:=(xi,eta)->A*eta;
```

$$f := (\xi, \eta) \mapsto A \cdot \eta \tag{9}$$

```
> m4:=y(tn+h)=m3;
```

$$m4 := y(tn+h) = (1-a1-a2)\,y(tn) + a1\,y(tn-h) + a2\,y(tn-2\,h) + h\left(\left(\frac{3}{8} - \frac{a1}{24}\right) A\,y(tn+h) + \left(\frac{19}{24} + \frac{13\,a1}{24} + \frac{a2}{3}\right) A\,y(tn) + \left(-\frac{5}{24} + \frac{13\,a1}{24} + \frac{4\,a2}{3}\right) A\,y(tn-h) + \left(\frac{1}{24} - \frac{a1}{24} + \frac{a2}{3}\right) A\,y(tn-2\,h)\right) \tag{10}$$

```
> ceq:=eval(subs(y=(s->rho^s),m4));
```

$$ceq := \rho^{tn+h} = (1-a1-a2)\,\rho^{tn} + a1\,\rho^{tn-h} + a2\,\rho^{tn-2\,h} + h\left(\left(\frac{3}{8} - \frac{a1}{24}\right) A\,\rho^{tn+h} + \left(\frac{19}{24} + \frac{13\,a1}{24} + \frac{a2}{3}\right) A\,\rho^{tn} + \left(-\frac{5}{24} + \frac{13\,a1}{24} + \frac{4\,a2}{3}\right) A\,\rho^{tn-h} + \left(\frac{1}{24} - \frac{a1}{24} + \frac{a2}{3}\right) A\,\rho^{tn-2\,h}\right) \tag{11}$$

```
> ceq2:=subs({a1=0,a2=2/5,tn=1,h=1},ceq);
```

$$ceq2 := \rho^2 = \frac{3\,\rho}{5} + \frac{2}{5\,\rho} + \frac{3\,A\,\rho^2}{8} + \frac{37\,A\,\rho}{40} + \frac{13\,A}{40} + \frac{7\,A}{40\,\rho} \tag{12}$$

```
> S2:=solve(ceq2,rho):
> Z1:=subs(A=a+I*b,abs(S2[1])):
> Z2:=subs(A=a+I*b,abs(S2[2])):
> Z3:=subs(A=a+I*b,abs(S2[3])):
> with(plots):
> contourplot(max(Z1,Z2,Z3),a=-6..1,b=-2..2,contours=[1],
    grid=[100,100],filled=true);
```




```
> cor1:=subs(S1,subs(E5=0,ynp1));
```

$$cor1 := (1 - a1 - a2)\, y(tn) + a1\, y(tn - h) + a2\, y(tn - 2\,h) + h\left(\left(\frac{3}{8} - \frac{a1}{24}\right) D(y)(tn + h) + \left(\frac{19}{24} + \frac{13\,a1}{24} + \frac{a2}{3}\right) D(y)(tn) + \left(-\frac{5}{24} + \frac{13\,a1}{24} + \frac{4\,a2}{3}\right) D(y)(tn - h) + \left(\frac{1}{24} - \frac{a1}{24} + \frac{a2}{3}\right) D(y)(tn - 2\,h)\right) \quad \textbf{(13)}$$